Problem #1) Slider on incline

(see hand written)

(see hand written)

Problem #2) Moment of Inertia

In[1]:=
```
Clear["Global`*"]
```

In[2]:=
```
location = {
    {1, 0, 0},
    {0, 1, 2},
    {0, 2, 1},
    {3, 1, 0},
    {1, 3, 0}
  };
mass = {1, 2, 2, 3, 3};
v = {vx, vy, vz} = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
nMax = 5 (* Number of masses *)
```

Out[5]= 5

Note, we program diagonal and off - diagonal with different formulas below!!!

In[6]:=
```
Clear[term]
term[n_, i_, i_] := +mass[[n]] (location[[n]].location[[n]] - (location[[n]].v[[i]]) ^ 2)
term[n_, i_, j_] := -mass[[n]] (location[[n]].v[[i]] * location[[n]].v[[j]])
```

In[9]:=
```
mat = Table[Sum[term[n, i, j], {n, 1, nMax}], {i, 1, 3}, {j, 1, 3}];
mat // MatrixForm
```

Out[10]//MatrixForm=

$$\begin{pmatrix} 50 & -18 & 0 \\ -18 & 41 & -8 \\ 0 & -8 & 71 \end{pmatrix}$$

Problem #3) Rotation Matrices

$$M_1 = \begin{pmatrix} \frac{1}{2} & +\frac{\sqrt{3}}{2} \\ -\frac{\sqrt{3}}{2} & \frac{1}{2} \end{pmatrix} \quad M_2 = \begin{pmatrix} \frac{1}{2} & +\frac{\sqrt{3}}{2} \\ +\frac{\sqrt{3}}{2} & \frac{1}{2} \end{pmatrix} \quad M_3 = \begin{pmatrix} -1 & 0 \\ 0 & -1 \end{pmatrix}$$

$$M_4 = \begin{pmatrix} -1 & 0 & \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{pmatrix} \quad M_5 = \begin{pmatrix} 1 & 0 \\ 0 & 3 \end{pmatrix} \quad M_6 = \begin{pmatrix} 2 & 0 \\ 0 & \frac{1}{2} \end{pmatrix}$$

ANSWERS :

M1: Yes, rotation by $\theta=\pi/3$

M2: No, need one negative and one positive Sinθ term

M3: Yes, rotation by 180 Degrees

M4: No, this flips all axes. (This is a parity transformation.)

M5: No, this will stretch the y-axis.

M6: No, this will stretch x-axis and compress y-axis

Problem #4) EigenValues and EigenVectors

In[11]:=
```
Clear["Global`*"]
```

In[12]:=
$$m = \begin{pmatrix} 5 & 0 & 0 \\ 0 & \frac{27}{3} & \frac{8}{3} \\ 0 & \frac{18}{3} & \frac{27}{3} \end{pmatrix};$$

In[13]:=
```
(* Just a cross check on the below *)
Eigenvalues[m]
```

Out[13]= {13, 5, 5}

In[14]:=
```
(* Just a cross check on the below *)
Eigenvectors[m]
```

Out[14]= $\left\{\left\{0, \frac{2}{3}, 1\right\}, \left\{0, -\frac{2}{3}, 1\right\}, \{1, 0, 0\}\right\}$

In[15]:=
```
one = DiagonalMatrix[{1, 1, 1}];
Det[m - λ one]
```

Out[16]= $325 - 155\,\lambda + 23\,\lambda^2 - \lambda^3$

In[17]:=
```
sol = Solve[Det[m - λ one] == 0, λ]
```
Out[17]= {{λ → 5}, {λ → 5}, {λ → 13}}

In[18]:=
```
eqs = (m - λ one).{a, b, c} == 0 // Thread
```
Out[18]= $\left\{a (5-\lambda) == 0, \frac{8 c}{3} + b (9-\lambda) == 0, 6 b + c (9-\lambda) == 0\right\}$

In[19]:=
```
sol[[1]]
```
Out[19]= {λ → 5}

In[20]:=
```
sol1 = Solve[ eqs /. sol[[1]], {a, b, c}][[1]]
```
••• Solve : Equations may not give solutions for all "solve " variables .

Out[20]= $\left\{c \to -\frac{3 b}{2}\right\}$

In[21]:=
```
v1 = {a, b, c} /. sol1
```
Out[21]= $\left\{a, b, -\frac{3 b}{2}\right\}$

In[22]:=
```
v1 = {a, b, c} /. sol1 /. {a → 1, b → 0}
```
Out[22]= {1, 0, 0}

In[23]:=
```
m.v1 == 5 v1
```
Out[23]= True

In[24]:=
```
v2 = {a, b, c} /. sol1 /. {a → 0, b → -1}
```
Out[24]= $\left\{0, -1, \frac{3}{2}\right\}$

In[25]:=
```
m.v2 == 5 v2
```
Out[25]= True

In[26]:=
```
sol[[3]]
```
Out[26]= {λ → 13}

In[27]:=
```
sol3 = Solve[ eqs /. sol[[3]], {a, b, c}][[1]]
```
••• Solve : Equations may not give solutions for all "solve " variables .

Out[27]= $\left\{a \to 0, c \to \frac{3 b}{2}\right\}$

In[28]:=
```
v3 = {a, b, c} /. sol3 /. {b → 1}
```
Out[28]= $\left\{0, 1, \frac{3}{2}\right\}$

In[29]:=
```
m.v1 == 5 v1
```
Out[29]=
```
True
```

In[30]:=
```
m.v2 == 5 v2
```
Out[30]=
```
True
```

In[31]:=
```
m.v3 == 13 v3
```
Out[31]=
```
True
```

Problem #5) Tides

In[32]:=
```
Clear["Global`*"]
```

In[33]:=
```
distMoon = 384 000 × 10^3;
distSun = 150 × 10^6 × 10^3;
mSun = 2 × 10^30;
mMoon = 7.3 × 10^22 ;
g = 6.67 × 10^-11;
m0 = 1;
```

Part a)

In[39]:=

$$fmoon = \frac{g\, m0\, mMoon}{distMoon^2};$$

```
fmoon // ScientificForm
```

Out[40]//ScientificForm=

3.30207×10^{-5}

In[41]:=

$$fsun = \frac{g\, m0\, mSun}{distSun^2};$$

```
fsun // ScientificForm
```

Out[42]//ScientificForm=

5.92889×10^{-3}

In[43]:=

$$\frac{fsun}{fmoon}$$

Out[43]= 179.551

Part b) Let’s look at the difference of the forces on opposite sides of the earth.

Note: this is extra and was not part of the required problem.
The key idea is the DIFFERENCE between the front and back face of the earth;

hence, the SLOPE of the force.

In[44]:=
```
rEarth = 6400 × 10^3;
```

In[45]:=
$$fmoonNear = \frac{g\ m0\ mMoon}{(distMoon - rEarth)^2};$$

$$fmoonFar = \frac{g\ m0\ mMoon}{(distMoon + rEarth)^2};$$

$$\frac{fmoonNear}{fmoonFar}$$

Out[47]= 1.06895

In[48]:=
$$fsunNear = \frac{g\ m0\ mSun}{(distSun - rEarth)^2};$$

$$fsunFar = \frac{g\ m0\ mSun}{(distSun + rEarth)^2};$$

$$\frac{fsunNear}{fsunFar}$$

Out[50]= 1.00017

Putting things on the same scale, the difference between the ffront and back of the earth is about 7 percent for the moon, and TINY for the sun!!!

In[51]:=
```
p1 = Plot[(g m0 mMoon / r) / (g m0 mMoon / distMoon),
  {r, distMoon - rEarth, distMoon + rEarth}, AxesOrigin → {Automatic, 0.95},
  PlotStyle → Blue,
  Ticks → {{{distMoon - 0.8 rEarth, "Near Side"},
     {distMoon, "Center"}, {distMoon + 0.9 rEarth, "Far Side"}}, Automatic}
 ]
```

Out[51]=



In[52]:=
```
p2 = Plot[(g m0 mSun / r) / (g m0 mSun / distSun), {r, distSun - rEarth, distSun + rEarth},
  AxesOrigin → {Automatic, 0.95}, PlotStyle → Red,
  Ticks → {{{distSun - 0.8 rEarth, "Near Side"}, {distSun, "Center"},
     {distSun + 0.9 rEarth, "Far Side"}}, Automatic}]
```

Out[52]=



Problem #6) M and K matrices

In[53]:=
```
Clear["Global`*"]
```

In[54]:=
$$\text{Tmat} = \begin{pmatrix} m & 2m & 0 \\ 2m & m & 0 \\ 0 & 0 & m \end{pmatrix};$$

```
Tmat // MatrixForm
```

Out[55]//MatrixForm=
$$\begin{pmatrix} m & 2m & 0 \\ 2m & m & 0 \\ 0 & 0 & m \end{pmatrix}$$

In[56]:=
```
Vmat = k {{2, 1, -1}, {1, 2, 1}, {-1, 1, 2}};
Vmat // MatrixForm
```

Out[57]//MatrixForm=
$$\begin{pmatrix} 2k & k & -k \\ k & 2k & k \\ -k & k & 2k \end{pmatrix}$$

In[58]:=
```
mat = Vmat - Tmat ω2;
mat // MatrixForm
```

Out[59]//MatrixForm=
$$\begin{pmatrix} 2k - m\,\omega 2 & k - 2m\,\omega 2 & -k \\ k - 2m\,\omega 2 & 2k - m\,\omega 2 & k \\ -k & k & 2k - m\,\omega 2 \end{pmatrix}$$

In[60]:=
```
Det[mat] /. {k → 1, m → 1} // Expand
```

Out[60]= $3\,\omega 2 - 6\,\omega 2^2 + 3\,\omega 2^3$

In[61]:=
```
Det[mat] // Factor
```

Out[61]= $3\,m\,\omega 2\,(k - m\,\omega 2)^2$

In[62]:=
```
sol = Solve[Det[mat] == 0, ω2] // Simplify
```

Out[62]= $\left\{\{\omega 2 \to 0\},\ \left\{\omega 2 \to \frac{k}{m}\right\},\ \left\{\omega 2 \to \frac{k}{m}\right\}\right\}$

In[63]:=
```
eq1 = mat .{a, b, c} == 0 // Thread;
eq1 // Column
```

Out[64]=
$-c\,k + b\,(k - 2\,m\,\omega 2) + a\,(2\,k - m\,\omega 2) == 0$
$c\,k + a\,(k - 2\,m\,\omega 2) + b\,(2\,k - m\,\omega 2) == 0$
$-a\,k + b\,k + c\,(2\,k - m\,\omega 2) == 0$

In[65]:=
```
sol1 = Solve[eq1 /. sol[[1]], {a, b, c}][[1]] // Simplify
```

··· Solve : Equations may not give solutions for all "solve " variables .

Out[65]= $\{b \to -a,\ c \to a\}$

In[66]:=
```
ev1 = {a, b, c} /. sol1 /. {a → 1} // Simplify
```
Out[66]= {1, -1, 1}

In[67]:=
```
mat.ev1 /. sol[[1]]
```
Out[67]= {0, 0, 0}

In[68]:=
```
sol2 = Solve[eq1 /. sol[[2]], {a, b, c}][[1]]  // Simplify  // PowerExpand
```
••• Solve : Equations may not give solutions for all "solve " variables .

Out[68]= {c → a - b}

In[69]:=
```
ev2 = {a, b, c} /. sol2 /. {a → 1, b → 1} // Simplify
```
Out[69]= {1, 1, 0}

In[70]:=
```
mat.ev2 /. sol[[2]]
```
Out[70]= {0, 0, 0}

In[71]:=
```
ev3 = {a, b, c} /. sol2 /. {a → -1, b → 1} // Simplify
```
Out[71]= {-1, 1, -2}

In[72]:=
```
mat.ev3 /. sol[[3]]
```
Out[72]= {0, 0, 0}

In[73]:=
```
eVecs = Normalize /@ {ev1, ev2, ev3} // Simplify
```
Out[73]= $\left\{\left\{\frac{1}{\sqrt{3}}, -\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right\}, \left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0\right\}, \left\{-\frac{1}{\sqrt{6}}, \frac{1}{\sqrt{6}}, -\sqrt{\frac{2}{3}}\right\}\right\}$

In[74]:=
```
Vdiag = eVecs.Vmat.Transpose[eVecs]  // FullSimplify ;
Vdiag // MatrixForm
```
Out[75]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 3\,k & 0 \\ 0 & 0 & \frac{k}{3} \end{pmatrix}$$

In[76]:=
```
Tdiag = eVecs.Tmat.Transpose[eVecs]  // FullSimplify ;
Tdiag  // MatrixForm
```
Out[77]//MatrixForm=

$$\begin{pmatrix} -\frac{m}{3} & 0 & 0 \\ 0 & 3\,m & 0 \\ 0 & 0 & \frac{m}{3} \end{pmatrix}$$

Part B Look at motion

Note: this is extra and was not part of the required problem.

In[78]:=
```
values = {m → 1, k → 1};
```

In[79]:=
```
sol /. values
```
Out[79]=
```
{{ω2 → 0}, {ω2 → 1}, {ω2 → 1}}
```

In[80]:=
```
mode1 = ev1 Exp[I ω t] /. {ω → Sqrt[ω2]} /. sol[[1]] /. values
```
Out[80]=
```
{1, -1, 1}
```

In[81]:=
```
mode1 = ev1 (x0 + v0 t) /. {ω → Sqrt[ω2]} /. sol[[1]] /. values /. {x0 → 0, v0 → 1}
```
Out[81]=
```
{t, -t, t}
```

In[82]:=
```
mode2 = ev2 Exp[I ω t] /. {ω → Sqrt[ω2]} /. sol[[2]] /. values
```
Out[82]= $\{e^{i t}, e^{i t}, 0\}$

In[83]:=
```
mode3 = ev3 Exp[I ω t] /. {ω → Sqrt[ω2]} /. sol[[3]] /. values
```
Out[83]= $\{-e^{i t}, e^{i t}, -2 e^{i t}\}$

In[84]:=
```
mode1 + {-1, 0, 1} /. values
```
Out[84]=
```
{-1 + t, -t, 1 + t}
```

In[85]:=
```
Plot[mode1 + {-1, 0, 1} /. values // Re, {t, 0, 10}]
```
Out[85]=



In[86]:=
```
Plot[mode2 + {-2, 0, 2} /. values // Re, {t, 0, 10}]
```
Out[86]=



In[87]:=
```
Plot[mode3 + {-2, 0, 2} /. values // Re, {t, 0, 10}]
```

Out[87]=



Problem #7) Mass and pulley

In[88]:=
```
Clear["Global`*"]
```

Problem 1: Mass and pulley



In[89]:=
```
Clear["Global`*"]
```

In[90]:= $\text{Tenergy} = \frac{1}{2}\,\text{m1}\,\text{x1}'[t]^2 + \frac{1}{2}\,\text{m2}\,\text{x2}'[t]^2 + \frac{1}{2}\,\text{Inertia}\,\theta'[t]^2\ /.\ \{\text{Inertia} \to \kappa\,\text{m3}\,r^2\}$

Out[90]= $\frac{1}{2}\,\text{m1}\,\text{x1}'[t]^2 + \frac{1}{2}\,\text{m2}\,\text{x2}'[t]^2 + \frac{1}{2}\,\text{m3}\,r^2\,\kappa\,\theta'[t]^2$

In[91]:=
```
Vpot = -m1 g x1[t] Sin[α] - m2 g x2[t];
```

Without Lagrange multipliers [as a cross check]

Note: this is extra and was not part of the required problem.

In[92]:=
```
constraint1 = x1[t] + x2[t] - len == 0;
dsol1 = DSolve[constraint1 , x2, t][[1]]
```

Out[93]= {x2 → Function[{t}, len - x1[t]]}

In[94]:=
```
constraint2 = r θ[t] - x1[t] == 0;
dsol2 = DSolve[constraint2 , θ, t][[1]]
```

Out[95]= $\left\{\theta \to \text{Function}\left[\{t\}, \frac{\text{x1}[t]}{r}\right]\right\}$

In[96]:=
```
Lag = Tenergy - Vpot /. dsol1 /. dsol2
```

Out[96]= $g\ m2\ (\text{len} - \text{x1}[t]) + g\ m1\ \text{Sin}[\alpha]\ \text{x1}[t] + \frac{1}{2}\ m1\ \text{x1}'[t]^2 + \frac{1}{2}\ m2\ \text{x1}'[t]^2 + \frac{1}{2}\ m3\ \kappa\ \text{x1}'[t]^2$

In[97]:=
```
Variables[Lag]
```

Out[97]= {g, len, m1, m2, m3, κ, Sin[α], x1[t], x1′[t]}

In[98]:=
```
eq1 = D[D[Lag, x1'[t]], t] - D[Lag, x1[t]] == 0
```

Out[98]= $g\ m2 - g\ m1\ \text{Sin}[\alpha] + m1\ \text{x1}''[t] + m2\ \text{x1}''[t] + m3\ \kappa\ \text{x1}''[t] == 0$

In[99]:=
```
check = Solve[eq1, x1''[t]][[1]] // Simplify
```

Out[99]= $\left\{\text{x1}''[t] \to \frac{g\ (-m2 + m1\ \text{Sin}[\alpha])}{m1 + m2 + m3\ \kappa}\right\}$

With Lagrange multipliers

In[100]:=
```
constraint1 = x1[t] + x2[t] - len == 0;
constraint1 [[1]]
```

Out[101]= -len + x1[t] + x2[t]

In[102]:=
```
constraint2 = r θ[t] - x1[t] == 0;
constraint2 [[1]]
```

Out[103]= -x1[t] + r θ[t]

In[104]:=
```
qlist1 = λ1 D[constraint1 [[1]], #] & /@ {x1[t], x2[t], θ[t]}
```

Out[104]= {λ1, λ1, 0}

In[105]:=
```
qlist2 = λ2 D[constraint2 [[1]], #] & /@ {x1[t], x2[t], θ[t]}
```

Out[105]= {-λ2, 0, r λ2}

In[106]:=
```
qlist = qlist1 + qlist2
```

Out[106]= {λ1 - λ2, λ1, r λ2}

In[107]:= **Lag = Tenergy - Vpot**

Out[107]= $g\,m1\,\text{Sin}[\alpha]\,x1[t] + g\,m2\,x2[t] + \frac{1}{2}\,m1\,x1'[t]^2 + \frac{1}{2}\,m2\,x2'[t]^2 + \frac{1}{2}\,m3\,r^2\,\kappa\,\theta'[t]^2$

In[108]:= **Variables[Lag]**

Out[108]= {g, m1, m2, m3, r, κ, Sin[α], x1[t], x2[t], x1′[t], x2′[t], θ′[t]}

In[109]:= **eq1 = D[D[Lag, x1'[t]], t] - D[Lag, x1[t]] == qlist[[1]]**

Out[109]= $-g\,m1\,\text{Sin}[\alpha] + m1\,x1''[t] == \lambda 1 - \lambda 2$

In[110]:= **eq2 = D[D[Lag, x2'[t]], t] - D[Lag, x2[t]] == qlist[[2]]**

Out[110]= $-g\,m2 + m2\,x2''[t] == \lambda 1$

In[111]:= **eq3 = D[D[Lag, θ'[t]], t] - D[Lag, θ[t]] == qlist[[3]]**

Out[111]= $m3\,r^2\,\kappa\,\theta''[t] == r\,\lambda 2$

In[112]:= **eq4 = D[constraint1, {t, 2}]**

Out[112]= $x1''[t] + x2''[t] == 0$

In[113]:= **x2sol = Solve[D[constraint1, {t, 2}], x2''[t]][[1]]**

Out[113]= $\{x2''[t] \to -x1''[t]\}$

In[114]:= **eq5 = D[constraint2, {t, 2}]**

Out[114]= $-x1''[t] + r\,\theta''[t] == 0$

In[115]:= **Solve[{eq1, eq2, eq3, eq4, eq5}, {x1''[t], x2''[t], θ''[t], λ1, λ2}] // Simplify**

Out[115]= $\left\{\left\{x1''[t] \to \frac{g\,(-m2 + m1\,\text{Sin}[\alpha])}{m1 + m2 + m3\,\kappa},\ x2''[t] \to \frac{g\,(m2 - m1\,\text{Sin}[\alpha])}{m1 + m2 + m3\,\kappa},\ \theta''[t] \to \frac{g\,(-m2 + m1\,\text{Sin}[\alpha])}{r\,(m1 + m2 + m3\,\kappa)},\ \lambda 1 \to -\frac{g\,m2\,(m1 + m3\,\kappa + m1\,\text{Sin}[\alpha])}{m1 + m2 + m3\,\kappa},\ \lambda 2 \to \frac{g\,m3\,\kappa\,(-m2 + m1\,\text{Sin}[\alpha])}{m1 + m2 + m3\,\kappa}\right\}\right\}$

In[116]:= **check**

Out[116]= $\left\{x1''[t] \to \frac{g\,(-m2 + m1\,\text{Sin}[\alpha])}{m1 + m2 + m3\,\kappa}\right\}$

Problem #8)Non-Intertial Frames

See handwritten for full solution



In[117]:=
```
Ω = Ω0 {0, 0, 1};
r = r0 {0, Sin[θ], Cos[θ]};
v = v0 {0, -Sin[θ], -Cos[θ]};
```

In[120]:=
```
values = {θ → 57 Degree, v0 -> 100, t → 100, Ω0 → 2 π/(24 × 60 × 60), r0 → 6400 × 10^3};
```

In[121]:=
```
(* Direction :east:   -x direction *)
coriolis = 2 m Cross[v, Ω]
```

Out[121]= {-2 m v0 Ω0 Sin[θ], 0, 0}

In[122]:=
```
(* East is negative x axis *)
eastForce = {-1, 0, 0}.coriolis
```

Out[122]= 2 m v0 Ω0 Sin[θ]

In[123]:=
```
(* Direction :   +y direction *)
centrifugal = m Cross[Cross[Ω, r], Ω]
```

Out[123]= {0, m r0 Ω0^2 Sin[θ], 0}

In[124]:=
```
(* Project component of centrifugal tangent to the earth's surface *)
```

In[125]:=
```
tangent = {0, Cos[θ], -Sin[θ]};
(* Check it is perpendicular to v (and r) and points south *)
{tangent.v, tangent.r}
```

Out[126]= {0, 0}

In[127]:=
```
southForce = centrifugal.tangent
```

Out[127]= $m\, r0\, \Omega 0^2 \, \text{Cos}[\theta]\, \text{Sin}[\theta]$

In[128]:= $\text{eastAcc} = \frac{\text{eastForce}}{m}$

Out[128]= $2\, v0\, \Omega 0\, \text{Sin}[\theta]$

In[129]:= $\text{southAcc} = \frac{\text{southForce}}{m}$

Out[129]= $r0\, \Omega 0^2\, \text{Cos}[\theta]\, \text{Sin}[\theta]$

In[130]:= (* $x=x0+v0\ t + \frac{1}{2} a t^2$ *)

$\text{southDisplacement} = \frac{1}{2}\, \text{southAcc}\ t^2$

Out[130]= $\frac{1}{2}\, r0\, t^2\, \Omega 0^2\, \text{Cos}[\theta]\, \text{Sin}[\theta]$

In[131]:=
```
southDisplacement //. values // N
```

Out[131]= 77.3005

In[132]:= (* $x=x0+v0\ t + \frac{1}{2} a t^2$ *)

$\text{eastDisplacement} = \frac{1}{2}\, \text{eastAcc}\ t^2$

Out[132]= $t^2\, v0\, \Omega 0\, \text{Sin}[\theta]$

In[133]:=
```
eastDisplacement //. values // N
```

Out[133]= 60.9898

Part c: let v be East (negative x direction)

In[134]:=
```
Ω = Ω0 {0, 0, 1};
r = r0 {0, Sin[θ], Cos[θ]};
v = v0 {-1, 0, 0};
```

In[137]:=
```
(* Direction : +y direction *)
coriolis = 2 m Cross[v, Ω]
```

Out[137]= {0, 2 m v0 Ω0, 0}

In[138]:=
```
(* Direction :   +y direction *)
centrifugal = m Cross[Cross[Ω, r], Ω]
```
Out[138]= $\{0,\ m\, r0\, \Omega0^2 \,\mathrm{Sin}[\theta],\ 0\}$

Problem #9) Without gravity

In[139]:=
```
Clear["Global`*"]
```

In[140]:= $T = \frac{1}{2} m1\, r^2\, \phi1'[t]^2 + \frac{1}{2} m2\, r^2\, \phi2'[t]^2;$

In[141]:= $V = \frac{1}{2} k\, (r\, \phi1[t] - r\, \phi2[t])^2;$

In[142]:=
```
lag = T - V
```
Out[142]= $-\frac{1}{2} k\, (r\, \phi1[t] - r\, \phi2[t])^2 + \frac{1}{2} m1\, r^2\, \phi1'[t]^2 + \frac{1}{2} m2\, r^2\, \phi2'[t]^2$

In[143]:=
```
D[D[lag, ϕ1'[t]], t] - D[lag, ϕ1[t]]  // Expand
```
Out[143]= $k\, r^2\, \phi1[t] - k\, r^2\, \phi2[t] + m1\, r^2\, \phi1''[t]$

In[144]:=
```
D[D[lag, ϕ2'[t]], t] - D[lag, ϕ2[t]]  // Expand
```
Out[144]= $-k\, r^2\, \phi1[t] + k\, r^2\, \phi2[t] + m2\, r^2\, \phi2''[t]$

In[145]:=
```
Tmat = {{m1 r^2, 0}, {0, m2 r^2}};
Tmat // MatrixForm
```
Out[146]//MatrixForm=
$$\begin{pmatrix} m1\, r^2 & 0 \\ 0 & m2\, r^2 \end{pmatrix}$$

In[147]:=
```
Vmat = k r^2 {{1, -1}, {-1, 1}};
Vmat // MatrixForm
```
Out[148]//MatrixForm=
$$\begin{pmatrix} k\, r^2 & -k\, r^2 \\ -k\, r^2 & k\, r^2 \end{pmatrix}$$

In[149]:=
```
mat = Vmat - Tmat ω2;
mat // MatrixForm
```
Out[150]//MatrixForm=
$$\begin{pmatrix} k\, r^2 - m1\, r^2\, \omega2 & -k\, r^2 \\ -k\, r^2 & k\, r^2 - m2\, r^2\, \omega2 \end{pmatrix}$$

In[151]:=
```
mrule = {m1 → m, m2 → m};
mat = mat /. mrule;
mat // MatrixForm
```

Out[153]//MatrixForm=

$$\begin{pmatrix} k\, r^2 - m\, r^2\, \omega 2 & -k\, r^2 \\ -k\, r^2 & k\, r^2 - m\, r^2\, \omega 2 \end{pmatrix}$$

In[154]:=
```
(* without gravity, the symmetric mode will just spin, so w=0 *)
sol = Solve[Det[mat] == 0, ω2] // Simplify
```

Out[154]= $\left\{\{\omega 2 \to 0\}, \left\{\omega 2 \to \frac{2\, k}{m}\right\}\right\}$

In[155]:=
```
eq1 = mat .{a, b} == 0 // Thread;
eq1 // Column
```

Out[156]=

$$-b\, k\, r^2 + a\, (k\, r^2 - m\, r^2\, \omega 2) == 0$$
$$-a\, k\, r^2 + b\, (k\, r^2 - m\, r^2\, \omega 2) == 0$$

In[157]:=
```
sol1 = Solve[eq1 /. sol[[1]], {a, b}][[1]] // Simplify
```

••• Solve : Equations may not give solutions for all "solve" variables .

Out[157]= {b → a}

In[158]:=
```
ev1 = {a, b} /. sol1 /. {a → 1} // Simplify
```

Out[158]= {1, 1}

In[159]:=
```
sol2 = Solve[eq1 /. sol[[2]], {a, b}][[1]] // Simplify // PowerExpand
```

••• Solve : Equations may not give solutions for all "solve" variables .

Out[159]= {b → -a}

In[160]:=
```
ev2 = {a, b} /. sol2 /. {a → 1} // Simplify
```

Out[160]= {1, -1}

In[161]:=
```
eVecs = Normalize /@ {ev1, ev2} // Simplify
```

Out[161]= $\left\{\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}\right\}$

In[162]:=
```
Vdiag = eVecs.Vmat.Transpose[eVecs] // FullSimplify ;
Vdiag // MatrixForm
```

Out[163]//MatrixForm=

$$\begin{pmatrix} 0 & 0 \\ 0 & 2\, k\, r^2 \end{pmatrix}$$

In[164]:=
```
Tdiag = eVecs.Tmat.Transpose[eVecs] /. mrule // FullSimplify ;
Tdiag // MatrixForm
```

Out[165]//MatrixForm=

$$\begin{pmatrix} m\,r^2 & 0 \\ 0 & m\,r^2 \end{pmatrix}$$

Part B Look at motion

Note: this is extra and was not part of the required problem.

In[166]:=
```
values = {m → 1, k → 1, r → 1};
```

In[167]:=
```
sol /. values
```

Out[167]= {{ω2 → 0}, {ω2 → 2}}

In[168]:=
```
mode1 = ev1 Exp[I ω t] /. {ω → Sqrt[ω2]} /. sol[[1]] /. values
```

Out[168]= {1, 1}

In[169]:=
```
mode1 = ev1 (x0 + v0 t) /. {ω → Sqrt[ω2]} /. sol[[1]] /. values /. {x0 → 0, v0 → 1}
```

Out[169]= {t, t}

In[170]:=
```
mode2 = ev2 Exp[I ω t] /. {ω → Sqrt[ω2]} /. sol[[2]] /. values
```

Out[170]= $\left\{e^{i\sqrt{2}\,t},\ -e^{i\sqrt{2}\,t}\right\}$

In[171]:=
```
Plot[mode1 + {-2, 2} /. values // Re, {t, 0, 10}]
```

Out[171]=



In[172]:=
```
Plot[mode2 + {-2, 2} /. values // Re, {t, 0, 10}]
```

Out[172]=



Problem #9) With gravity

Note: this is extra and was not part of the required problem.

In[173]:=
```
Clear["Global`*"]
```

In[174]:= $T = \frac{1}{2} m1\ r^2\ \phi1'[t]^2 + \frac{1}{2} m2\ r^2\ \phi2'[t]^2;$

In[175]:= $V = \frac{1}{2} k (r\ \phi1[t] - r\ \phi2[t])^2;$

In[176]:=
```
Vgravity1 = m1 g r (1 - Cos[φ1[t]]);
Vgravity2 = m2 g r (1 - Cos[φ2[t]]);
```

In[178]:=
```
(* Expand for small angles *)
Vgravity1 = Series[Vgravity1 , {φ1[t], 0, 2}] // Normal
```

Out[178]= $\frac{1}{2} g\ m1\ r\ \phi1[t]^2$

In[179]:=
```
(* Expand for small angles *)
Vgravity2 = Series[Vgravity2 , {φ2[t], 0, 2}] // Normal
```

Out[179]= $\frac{1}{2} g\ m2\ r\ \phi2[t]^2$

In[180]:=
```
lag = T - (V + Vgravity1 + Vgravity2)
```

Out[180]= $-\frac{1}{2} g\ m1\ r\ \phi1[t]^2 - \frac{1}{2} g\ m2\ r\ \phi2[t]^2 - \frac{1}{2} k (r\ \phi1[t] - r\ \phi2[t])^2 + \frac{1}{2} m1\ r^2\ \phi1'[t]^2 + \frac{1}{2} m2\ r^2\ \phi2'[t]^2$

In[181]:=
```
D[D[lag, φ1'[t]], t] - D[lag, φ1[t]] // Expand
```

Out[181]= $g\ m1\ r\ \phi1[t] + k\ r^2\ \phi1[t] - k\ r^2\ \phi2[t] + m1\ r^2\ \phi1''[t]$

In[182]:=
```
D[D[lag, ϕ2'[t]], t] - D[lag, ϕ2[t]] // Expand
```

Out[182]= $-k\, r^2\, \phi 1[t] + g\, m2\, r\, \phi 2[t] + k\, r^2\, \phi 2[t] + m2\, r^2\, \phi 2''[t]$

In[183]:=
```
Tmat = {{m1 r^2, 0}, {0, m2 r^2}};
Tmat // MatrixForm
```

Out[184]//MatrixForm=

$$\begin{pmatrix} m1\, r^2 & 0 \\ 0 & m2\, r^2 \end{pmatrix}$$

In[185]:=
```
Vmat = k r^2 {{1, -1}, {-1, 1}} + DiagonalMatrix[{m1 g r, m2 g r}];
Vmat // MatrixForm
```

Out[186]//MatrixForm=

$$\begin{pmatrix} g\, m1\, r + k\, r^2 & -k\, r^2 \\ -k\, r^2 & g\, m2\, r + k\, r^2 \end{pmatrix}$$

In[187]:=
```
mat = Vmat - Tmat ω2;
mat // MatrixForm
```

Out[188]//MatrixForm=

$$\begin{pmatrix} g\, m1\, r + k\, r^2 - m1\, r^2\, \omega 2 & -k\, r^2 \\ -k\, r^2 & g\, m2\, r + k\, r^2 - m2\, r^2\, \omega 2 \end{pmatrix}$$

In[189]:=
```
mrule = {m1 → m, m2 → m};
mat = mat /. mrule;
mat // MatrixForm
```

Out[191]//MatrixForm=

$$\begin{pmatrix} g\, m\, r + k\, r^2 - m\, r^2\, \omega 2 & -k\, r^2 \\ -k\, r^2 & g\, m\, r + k\, r^2 - m\, r^2\, \omega 2 \end{pmatrix}$$

In[192]:=
```
(* With gravity there is no zero frequency *)
sol = Solve[Det[mat] == 0, ω2] // Simplify
```

Out[192]= $\left\{\left\{\omega 2 \to \frac{g}{r}\right\}, \left\{\omega 2 \to \frac{2\,k}{m} + \frac{g}{r}\right\}\right\}$

In[193]:=
```
eq1 = mat.{a, b} == 0 // Thread;
eq1 // Column
```

Out[194]=

$-b\, k\, r^2 + a\,(g\, m\, r + k\, r^2 - m\, r^2\, \omega 2) == 0$

$-a\, k\, r^2 + b\,(g\, m\, r + k\, r^2 - m\, r^2\, \omega 2) == 0$

In[195]:=
```
sol1 = Solve[eq1 /. sol[[1]], {a, b}][[1]] // Simplify
```

Solve : Equations may not give solutions for all "solve " variables .

Out[195]= {b → a}

In[196]:=
```
ev1 = {a, b} /. sol1 /. {a → 1} // Simplify
```

Out[196]= {1, 1}

In[197]:=
```
sol2 = Solve[eq1 /. sol[[2]], {a, b}][[1]]  // Simplify  // PowerExpand
```
••• Solve : Equations may not give solutions for all "solve " variables .

Out[197]= $\{b \to -a\}$

In[198]:=
```
ev2 = {a, b} /. sol2  /. {a → 1} // Simplify
```
Out[198]= $\{1, -1\}$

In[199]:=
```
eVecs = Normalize /@ {ev1, ev2} // Simplify
```
Out[199]= $\left\{\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}\right\}$

In[200]:=
```
Vdiag = eVecs.Vmat.Transpose[eVecs]  /. mrule // FullSimplify ;
Vdiag // MatrixForm
```
Out[201]//MatrixForm=

$$\begin{pmatrix} g\,m\,r & 0 \\ 0 & r\,(g\,m + 2\,k\,r) \end{pmatrix}$$

In[202]:=
```
Tdiag = eVecs.Tmat.Transpose[eVecs]  /. mrule // FullSimplify ;
Tdiag  // MatrixForm
```
Out[203]//MatrixForm=

$$\begin{pmatrix} m\,r^2 & 0 \\ 0 & m\,r^2 \end{pmatrix}$$

Part B Look at motion

Note: this is extra and was not part of the required problem.

In[204]:=
```
values = {m → 1, k → 1, r → 1, g → 1};
```
In[205]:=
```
sol /. values
```
Out[205]= $\{\{\omega 2 \to 1\}, \{\omega 2 \to 3\}\}$

In[206]:=
```
mode1 = ev1 Exp[I ω t] /. {ω → Sqrt[ω2]} /. sol[[1]]  /. values
```
Out[206]= $\{e^{i\,t}, e^{i\,t}\}$

In[207]:=
```
mode2 = ev2 Exp[I ω t] /. {ω → Sqrt[ω2]} /. sol[[2]] /. values
```
Out[207]= $\{e^{i\sqrt{3}\,t}, -e^{i\sqrt{3}\,t}\}$

In[208]:= `Plot[mode1 + {-2, 2} /. values // Re, {t, 0, 10}]`

Out[208]=


In[209]:= `Plot[mode2 + {-2, 2} /. values // Re, {t, 0, 10}]`

Out[209]=


Problem #10) Relativity:

Part a)

In[210]:= $x = 0.5 \times 10^{-6}\ \mathrm{s}\ 3 \times 10^{8}\ \frac{\mathrm{m}}{\mathrm{s}}$

Out[210]= 150. m

Part b)

In[211]:= $\gamma = \frac{10 \times 10^{3}\ \mathrm{m}}{x}$

Out[211]= 66.6667

Part c)

In[212]:=
```
bSol = Solve[g == 1/Sqrt[1 - β^2], β][[2]]
```
$$\text{bSol} = \text{Solve}\left[g == \frac{1}{\sqrt{1-\beta^2}}, \beta\right][[2]]$$

Out[212]=
$$\left\{\beta \to \frac{\sqrt{-1+g^2}}{g}\right\}$$

In[213]:=
```
β /. bSol /. {g → γ}
```

Out[213]= 0.999887

Part d)

In[214]:=
$$\frac{\beta 1 + \beta 2}{1 + \beta 1\, \beta 2} \text{ // . } \{\beta 1 \to \beta 2, \beta 2 \to 0.8\}$$

Out[214]= 0.97561

Relativity: EIC

In[215]:=
$$\text{values} = \left\{p1 \to 275, p2 \to 18, mp \to 1, me \to \frac{1}{2000}\right\};$$

In[216]:=
```
g = DiagonalMatrix [{1, -1, -1, -1}];
```

Part e

In[217]:=
```
proton = {p1, 0, 0, p1};
electron = {p2, 0, 0, -p2};
p12 = proton + electron
```

Out[219]= {p1 + p2, 0, 0, p1 - p2}

In[220]:=
```
eq = s == p12.g.p12 // Simplify
```

Out[220]= 4 p1 p2 == s

In[221]:=
```
sol1 = Solve[eq, s][[1]]
```

Out[221]= {s → 4 p1 p2}

In[222]:=
```
Sqrt[s] /. sol1
```

Out[222]= $2\sqrt{p1\, p2}$

In[223]:=
```
Sqrt[s] /. sol1 /. values // N
```

Out[223]= 140.712

Part f

```
In[224]:= proton = {mp, 0, 0, 0};
          electron = {p2, 0, 0, -p2};
          p12 = proton + electron
Out[226]= {mp + p2, 0, 0, -p2}

In[227]:= eq = s == p12.g.p12 // Simplify
Out[227]= s == mp (mp + 2 p2)

In[228]:= sol1 = Solve[eq, s][[1]]
Out[228]= {s → mp (mp + 2 p2)}

In[229]:= Sqrt[s] /. sol1
Out[229]= √(mp (mp + 2 p2))

In[230]:= (* a smaller number *)
          Sqrt[s] /. sol1 /. values // N
Out[230]= 6.08276
```

Part g

```
In[231]:= proton = {p1, 0, 0, p1};
          electron = {me, 0, 0, 0};
          p12 = proton + electron
Out[233]= {me + p1, 0, 0, p1}

In[234]:= eq = s == p12.g.p12 // Simplify
Out[234]= s == me (me + 2 p1)

In[235]:= sol1 = Solve[eq, s][[1]]
Out[235]= {s → me (me + 2 p1)}

In[236]:= Sqrt[s] /. sol1
Out[236]= √(me (me + 2 p1))

In[237]:= (* a very small number *)
          Sqrt[s] /. sol1 /. values // N
Out[237]= 0.524405
```